Jörg Meyer, Hameln

Ein CAS-Weg zu Tangenten und zu Krümmungskreisen

1. Krümmungskreise

Man erhält den zu t gehörigen Krümmungskreismittelpunkt $M(t)$ der Funktion f, indem man die Normale zu $f(t)$ mit der Normalen zu $f(t+h)$ schneidet und im Schnittpunkt h gegen Null gehen lässt. Das Resultat ist $M(t)=\frac{1}{f''(t)}\cdot\begin{pmatrix} t\cdot f''(t)-f'(t)-(f'(t))^3 \\ f(t)\cdot f''(t)+1+(f'(t))^2 \end{pmatrix}$. Für $f(t)=t^2$ ist $M(t)=\begin{pmatrix} -4\cdot t^3 \\ \frac{1}{2}+3\cdot t^2 \end{pmatrix}$, für $f(t)=t^3-3\cdot t$ ist $M(t)=\frac{1}{6\cdot t}\cdot\begin{pmatrix} 3\cdot(10-26\cdot t^2+27\cdot t^4-9\cdot t^6) \\ 10-36\cdot t^2+15\cdot t^4 \end{pmatrix}$.

Aber man bekommt die Mittelpunkte der Krümmungskreise auch ganz anders:

Der Krümmungskreis an der Stelle t zeichnet sich dadurch aus, dass er $x=t$ als dreifache Schnittstelle mit der Kurve hat. Damit ist der Krümmungskreis eindeutig bestimmt.

Beispiel Normalparabel mit $y=x^2$: Ein beliebiger Kreis hat die Gleichung $(x-u)^2+(y-v)^2=R^2$. Die Schnittgleichung soll die dreifache Nullstelle t haben, d.h. der Rest bei der Polynomdivision $\frac{(x-u)^2+(x^2-v)^2-R^2}{(x-t)^3}$ muss verschwinden. Aus den Koeffizienten des Restes r ermittelt man u, v und sogar R! Hier der Maxima-Code:

```
t1: (x-u)^2+(x^2-v)^2-R^2$
t2:(x-t)^3$
r:remainder(t1,t2);
    -R^2+(-2*v+6*t^2+1)*x^2+(-2*u-8*t^3)*x+v^2+u^2+3*t^4
so_u:solve(coeff(r,x,1), u)[1];
    u=-4*t^3
so_v:solve(coeff(r,x,2) ,v)[1];
    v=(6*t^2+1)/2
so_R:solve(coeff(r,x,0) ,R);
    [R=-sqrt(v^2+u^2+3*t^4),R=sqrt(v^2+u^2+3*t^4)]
subst([u=rhs(so_u), v=rhs(so_v)], so_R[2]), radcan;
    R=(4*t^2+1)^(3/2)/2
```

Damit hat man vom zur Stelle t gehörigen Krümmungskreis dessen Mittelpunkt und ebenfalls dessen Radius R.

Das funktioniert auch für allgemeine Polynomfunktionen, etwa für $f(t)=t^3-3\cdot t$; hier muss man den ersten Wert von u noch abändern, da er zunächst von v abhängt:

```
t1: (x-u)^2+(x^3-3*x-v)^2-R^2$
t2:(x-t)^3$
r:remainder(t1,t2)$
so_u:solve(coeff(r,x,1), u)[1];
   u=(3*t^2+3)*v-12*t^5+24*t^3
so_v:solve(coeff(r,x,2) ,v)[1];
   v=(15*t^4-36*t^2+10)/(6*t)
so_u1:subst([v=rhs(so_v)], so_u), ratsimp;
   u=-(9*t^6-27*t^4+26*t^2-10)/(2*t)
```

Stets gibt es für jedes $t \neq 0$ einen einzigen Kreis, der mit der Kurve dort eine 3-fache Schnittstelle hat.

Analog zur Parabel hätte man auch den Radius des Krümmungskreises bekommen.

Naturgemäß stößt das CAS-Verfahren bei der Exponentialfunktion ($f(t)=e^t$) an seine Grenzen, da schon der Zähler der Polynomdivision eine Potenzreihe ist.

2. Tangenten

Die Vorgehensweise lässt sich auch für die Tangentenermittlung verwenden: Die Tangente an der Stelle t zeichnet sich dadurch aus, dass sie $x=t$ als doppelte Schnittstelle mit der Kurve hat. Damit ist die Tangente eindeutig bestimmt.

Beispiel Normalparabel mit $y=x^2$: Eine beliebige (nicht parallel zur Hochachse verlaufende) Gerade hat die Gleichung $y=m\cdot x+n$. Die Schnittgleichung soll die doppelte Nullstelle t haben, d.h. der Rest bei der Polynomdivision $\frac{x^2-(m\cdot x+n)}{(x-t)^2}$ muss verschwinden. Aus den Koeffizienten des Restes r ermittelt man m und n. Hier der Maxima-Code:

```
t1:x^2-(m*x+n)$
t2:(x-t)^2$
r:remainder(t1, t2);
     (2*t-m)*x-t^2-n
m:solve(coeff(r,x,1), m)[1];
     m=2*t
n:solve(coeff(r,x,0), n)[1];
     n=-t^2
```

Man überzeugt sich leicht davon, dass dieser Tangentenermittlungsalgorithmus auch für beliebige Polynomfunktionen zum richtigen Resultat führt.